


Паньгин Александр Викторович

## МОДЕЛЬ «ЭЛЕКТРОННАЯ БИБЛИОТЕКА» В EXCEL

*Любая деятельность предусматривает свое совершенство, ее совершенство же предусматривает иную деятельность.*



### ВВЕДЕНИЕ

Создадим модель работы небольшой городской библиотеки (или медиатеки) по учету выдачи книг читателям. Цель данной модели – организация «безбумажного абонемента» на основе изучения возможностей электронных таблиц.

Данная модель имеет преимущество перед используемой в настоящее время бумажной технологией в простоте создания различных запросов по статистической обработке данных учета движения книг. При широком взгляде на поставленную задачу видится работа библиотечного абонемента без самих сотрудников библиотеки (впрочем, при наличии обычного контролера). Для этого, вероятно, достаточно использование встроенных чипов в книгах и читательских билетах.

В учебных целях далее будет рассмотрена упрощенная модель с ограниченными функциями, которые, впрочем, возможно пополнять при практическом или учебно-игровом исполнении. Итак, приступим к созданию модели в электронной таблице Excel пакета MS Office. Предполагается, что не требуется создания пользовательских макросов (которые в целом упрощают исходную задачу).



### ПОСТАНОВКА ЗАДАЧИ

Познакомившись с работой библиотеки в качестве читателя, определимся с концептуальной моделью – выделим основные функции библиотечного абонемента:

- вести текущий учет движения книг для конкретного читателя по дате выдачи и дате возврата книг;
- отслеживать замену или списание книг в общем фонде;
- устанавливать список книг, выданных на текущий момент времени, и определять читателей, имеющих их «на руках»;
- определять списки текущей задолженности читателей и начисления пени в зависимости от стоимости книг и дней задолженности;
- предупреждать библиотекаря о предельных ограничениях суммарной стоимости выбранных книг отдельным читателем.



### ПОСТРОЕНИЕ МОДЕЛИ

Функционально весь собранный материал можно разбить по пяти разделам с условными названиями: **Фонд**, **Абонемент** (или выдача книг), **Возврат книг**, **Читатели**, **Служебный** (конфиденциальный раздел информации). Создадим листы книги

Excel (не путать с библиотечными книгами) с соответствующими названиями. Подробное изложение содержащегося в них данных поясняет формализацию исходной информации. На демонстрационных рисунках модели используемые формулы в ячейках приводятся в примечаниях к ячейкам или в строке формул.



**РАЗДЕЛ «ФОНД»**

Прежде всего библиотека должна иметь книжный фонд (в общем случае, под книгой подразумевается широкий ассортимент печатных и электронных носителей информации).

На пространстве вкладки **Фонд** определим столбцы данных (рис. 1).

**Столбец А** – Инвентарный номер книги, для простоты зададим его числом (в общем случае, поле содержит уникальную для данной книги строку знаков) и будем в дальнейшем называть номером книги (номера книг не отсортированы).

**Столбцы В, С, D** – Наименование книги, Авторы и Издательство, Год издания – представляют соответственно информацию, не требующую дополнительных пояснений.

**Столбец Е** – Исходная цена (книги) – информация, необходимая для определения «ущерба», нанесенного библиотеке в случае просроченного возврата или утери книги читателем.

**Столбец F** – Коэффициент «ценности» книги – он может быть малым в случае «зачитанности» книги или большим, например, для выравнивания «рыночной» цены в случае эквивалентной замены книгами «старых» изданий.

**Столбец G** – Стоимость (книги, текущая), определяется формулой

$$=ЕСЛИ(Ч(F12)>0;E12*F12;E12)$$

для ячейки G12. Для остальных ячеек подобная формула означает, что если числовой коэффициент в столбце F положительный (то есть задано число), то он умножается на цену книги.

**Столбец H** – Замена – отмечает замененные книги фонда, например: книга с инвентарным номером 7 заменена на книгу с инвентарным номером 11, а книга с номером 8 на данный момент времени заменена на 0 (то есть списана).

**Столбец I** – Строка абонемент – если ячейка не пустая, то она указывает для листа Excel с именем «Абонемент» номер строки, в которой отмечено, кому и когда выдана данная книга. Пояснение формулы для ячеек столбца I будет приведено ниже, когда познакомимся со всеми данными модели.

Номер выданной книги фонда отмечается красным цветом. Условное форматирование для ячеек столбца А (например, A2) показано на рис. 2 выбором меню **Формат → Условное форматирование**.

	A	B	C	D	E	F	G	H	I	J
1	Инвентарный номер книги	Наименование книги	Авторы, Издательство	Год издания	Исходная Цена	Козффицент	Стоимость	Замена	Строка абонемент	
2	3	Бесы	Достоевский Ф.М.		330.00		330.00		10	
3	4	Подросток	Достоевский Ф.М.		290.00		290.00		10	
4	5	Игрок	Достоевский Ф.М.		200.00		200.00		5	
5	6	Темные аллеи	Бунин И.А.		280.00	0.2	56.00		4	
6	7	Мать	Горький М.		50.00		50.00	11		
7	1	Война и мир т.1	Толстой Л.Н.		180.00		180.00			
8	2	Война и мир т.2	Толстой Л.Н.		190.00		190.00			
9	8	Воскресение	Толстой Л.Н.		100.00		100.00	0		
10	9	Холстомер	Толстой Л.Н.		30.00		30.00		7	
11	10	Лирика	Пастернак Б.Л.		70.00		70.00		5	
12	11	Избранные сочинения	Горький М.	1986	5.70	15	85.50			
13										
14										
15										

Рис. 1





Рис. 2

Присвоим имя **Фонд** набору ячеек **Фонд!\$A\$1:\$G\$65**. Выделим указанные ячейки, откроем меню **Вставка → Имя → Присвоить**, в поле «Имя» наберем слово «Фонд», в поле «Формула» будет отмечен выделенный набор ячеек, нажмем кнопку «ОК» или «Добавить». Аналогичным образом присвоим набору ячеек **Фонд!\$A\$1:\$A\$65** имя **Инв\_номер**.

*Замечание.* Отмеченный набор ограничивается условной строкой 65. При создании модели это делается специально для проверки правильности вводимых формул или их фрагментов. В готовом приложении во всех именах число 65 следует заменить на большее число строчек Excel, например 65535. Ячейки с формулами не требуют рутинного «ручного» заполнения данными.



### РАЗДЕЛ «ЧИТАТЕЛИ»

Лист «Читатели» содержит сведения о читателях библиотеки (рис. 3):

- номер читательского билета для идентификации читателя библиотеки;
- дату регистрации билета;
- фамилию, имя, отчество читателя;

	А	В	С	Д	Е	Ф	Г
1	Читательский билет	Дата регистрации	ФИО	Документ	Дата рождения	Адрес	Номер телефона
2	7	03.03.2007	А А А	паспорт	16.05.1990	Адр7	47615
3	10	28.02.2007	Б Б Б	студ. билет		Адр10	
4	11	20.01.2007	В В В	паспорт		Адр11	20202
5	20	21.01.2007	С С С			Адр11	20202
6	13	21.02.2007	Д Д Д			Адр13	
7	27	26.02.2005	Х Х Х			Адр27	
8	34	05.03.2007	У У У			Адр34	71234

Рис. 3


– данные документа, удостоверяющие личность;

– адрес проживания;

Поля «Адрес» и «Номер телефона» необходимы для извещения читателю о его задолженности и предупреждении о начислении пени без уважительной причины

при возврате книг.

Поле «Дата рождения» присутствует для статистического анализа возрастных групп читателей.




### РАЗДЕЛ «СЛУЖЕБНЫЙ»

«Служебный» лист информации (рис. 4) содержит: в ячейке А2 – значение ставки пени (в процентах) и в ячейке В2 – дату (текущего дня), от которой производится поиск должников.

Ячейка В2 содержит формулу

=СЕГОДНЯ(),

дата текущего дня обновляется при открытии данного приложения Excel (и библиотеки).



### РАЗДЕЛ «АБОНЕМЕНТ»

Лист «Абонемент» имеет более сложный вид (рис. 5) и предназначен для хранения информации о выдаче книг.

**Столбец А** – Номер читательского билета (ячейка заполняется при выдаче книг).

**Столбец В** – Дата (и время) выдачи (книг), задается формулой, например для ячейки В10

=ЕСЛИ(А10<>"";ТДАТА();"").

	А	В
1	Ставка пени %	Отчет даты задолженности
2	0.01	06.04.2007
3		=СЕГОДНЯ()
4		

Рис. 4

Таким образом, при заполнении ячейки A10 формируется текущая дата в ячейке B10. Формат ячейки ДД.ММ.ГГГГ ч:мм

**Столбец С** – Дата возврата (книг), формируется формулой (для 10-й строки)

=ЕСЛИ(B10<>"";ДАТАМЕС(B10;1);""),

то есть рекомендуется сдать книги не позже одного месяца со дня выдачи (как увидим далее, при задержке будут начисляться пени).

**Столбцы D, E, F** – содержат инвентарные номера выдаваемых книг, соответственно для Книги1, Книги2, Книги3 (не более трех книг на одного читателя).

**Столбцы G, H, I** – Стоимости Книги1, Книги2, Книги3 соответственно, значение в ячейке G10 задается формулой

=ЕСЛИ(Ч(D10)>0;ВПР(D10;Фонд;7;ЛОЖЬ);""),

что означает, если выдана Книга1 с номером в ячейке D10, то производится поиск этого номера в крайнем левом столбце массива ячеек с именем **Фонд** и в найденной строке выбирается значение в ячейке 7-го столбца массива **Фонд** (стоимость книги). Параметр интервального просмотра ЛОЖЬ функции ВПР указывает, что поиск может производиться в неотсортированном массиве номеров книг фонда.

Для скрытия вспомогательных столбцов стоимостей отдельных книг выделите столбцы G, H, I, затем выберите команду меню **Данные → Группа и структура → Группировать**. Над выбранными столбца-

ми появится полоса уровня структуры и символ скрытия деталей (квадратик с «минусом»). Щелкая мышью на полосе уровня или символу скрытия деталей (или показа – квадратик с «плюсом»), можно скрывать (за ненадобностью) или раскрывать (по необходимости) соответствующие деталиные элементы модели.

**Столбец J** – Стоимость книг, подсчет суммарной стоимости выданных книг для конкретного читателя. Используем условное форматирование (рис. 6) фона ячейки для значений стоимости, которые выше критически принимаемой (условно, более 500).

**Столбец K** – Строка возврата, указывает номер строки листа «Возврат книг» в случае, если существует соответствующая информация о факте возврата данных книг читателем. Эта информация позволяет оперативно найти дату и другие обстоятельства возврата, а также использовать их для определения списка читателей-должников, для рассылки им (по почте или телефону) сообщений о задолженности.

**Столбец L** – Долг, идентифицирует буквой «д» тех читателей, у которых срок возврата книг (ячейка столбца С) назначен ранее, чем текущий день (в общем случае, ранее, чем установленное значение в ячейке B2 на листе «Служебный»), и взятые книги они не возвращали (ячейка поля «Строка возврата» пустая (а точ-

1	A	B	C	D	E	F	G	H	I	J	K	L
2	Читательский билет	Дата выдачи	Дата возврата	Книга1	Книга2	Книга3	Стоимость Книга1	Стоимость Книга2	Стоимость Книга3	Стоимость книг	Строка возврата	Долг
2	11	21.01.2007 9:00	21.02.2007	8	9	10	100.00	30.00	70.00	200.00	2	
3	20	21.01.2007 10:00	21.02.2007	1	2		180.00	190.00		370.00	3	
4	11	19.02.2007 10:30	19.03.2007	6			56.00			56.00		д
5	13	21.02.2007 13:00	21.03.2007	10		5	70.00		200.00	270.00		д
6	20	26.02.2007 14:00	26.03.2007	2	3		190.00	330.00		520.00	5	
7	10	28.02.2007 18:25	28.03.2007	9			30.00			30.00		д
8	7	03.03.2007 12:00	03.04.2007	4			290.00			290.00		4
9	34	05.03.2007 17:10	05.04.2007	1	8	7	180.00	100.00	50.00	330.00		6
10	11	05.04.2007 9:00	05.05.2007	4	3		290.00	330.00		620.00		
11	=ЕСЛИ(A10<>"";ТДАТА();"")											
12	=ЕСЛИ(B10<>"";ДАТАМЕС(B10;1);"")											
13	=ЕСЛИ(Ч(D10)>0;ВПР(D10;Фонд;7;ЛОЖЬ);"")											
14	=ЕСЛИ(И(С7<Служебный!\$B\$2; К7=0);"д";"")											
15	=СУММ(G10:I10)											

Рис. 5




Рис. 6

нее, имеет значение 0). Соответствующая формула для L7 ячейки

=ЕСЛИ(И(С7<Служебный!\$B\$2;K7=0);"д";"").



**РАЗДЕЛ «ВОЗВРАТ КНИГ»**

Лист возврата книг содержит следующие столбцы (рис. 7).

**Столбец А** – номер читательского билета.

**Столбец В** – дата фактического возврата книг, задается формулой, аналогичной полю «Дата выдачи» в разделе «Абонемент».

Для использования в формулах массивов ячеек для ясности определим их имена аналогично ранее присвоенному имени Фонд. Полный перечень имен указан в таблице 1.

Индекс А в имени относится к листу «Абонемент», индекс В – к листу «Возврат книг».

Вернемся к описанию раздела «Возврат книг» и ячеек с формулами.

**Столбец С** – Строка листа абонемента – содержит в ячейках (например, для С2) табличную формулу:

{=МАКС((Читатель\_А=A2)\*СТРОКА(Читатель\_А)(Дата\_выдачи<B2))}

Данной формулой производится поиск номера строки на листе «Абонемент», в которой содержится последняя (по дате) информация о выдаче книг данному читателю.

Опишем формулу по частям.

Фрагмент формулы (Читатель\_А=A2) возвращает массив логических значений ИСТИНА (1) и ЛОЖЬ (0), в зависимости от того, содержится ли номер читательского билета (содержимое А2) в строке массива «Читатель\_А». Если это значение – ЛОЖЬ, то фрагмент

(Читатель\_А=A2)\*СТРОКА(Читатель\_А)

возвращает значение 0, а если ИСТИНА – значение, равное номеру строки. Умножение на фрагмент (Дата\_выдачи<B2) селектирует ненулевые строки, для которых дата выдачи меньше даты возврата (в ячейке В2). Наконец, функция МАКС определяет максимальную строку с информацией о последней выдаче перед данным возвратом книг. Ввод табличной формулы должен завершаться совместным нажатием клавиш <Ctrl>+<Shift>+<Enter>.

**Столбец С** – (количество) Дней невозврата, определяется формулой, например для С6 ячейки

	А	В	С	Д	Е	Ф	Г	Н
1	Читательский билет	Дата (факт.) возврата	Строка листа абонемента	Дней невозврата	Пени			
2	11	19.02.07 9:50	2	0	0.00			
3	20	26.02.07 13:00	3	5	14.50			
4	7	03.03.07 12:30	8	0	0.00			
5	20	25.03.07 10:00	6	0	0.00			
6	34	06.04.07 9:15	9	1	1.00			
7								
8								
9								
10								
11								

=ЕСЛИ(А6<>"";ТДАТА0;"")

=ИНДЕКС(Стоимость;С6)\*D6\*Служебный!\$A\$2

=ЕСЛИ(В6>ИНДЕКС(Дата\_возврата\_А;С6); ЦЕЛОЕ(В6-ИНДЕКС(Дата\_возврата\_А;С6));0)

Рис. 7

=ЕСЛИ(В6>ИНДЕКС(Дата\_возврата\_А;С6); ЦЕЛОЕ(В6-ИНДЕКС(Дата\_возврата\_А; С6));0)

То есть если дата фактического возврата больше даты запланированного возврата (которая расположена на листе «Абонемент» в строке с найденным номером в ячейке С6), то будет возвращено целое значение от разницы дат, иначе – значение 0. В ячейках столбца С задать формат «Общий».

**Столбец Е** – Пени, осуществляет подсчет пени перемножением значений: суммарной стоимости выданных книг (в столбце J листа «Абонемент»), количества дней невозврата, процентной ставки пени.

*Замечания*

1. При отладке формулы можно в строке формулы увидеть значения, которые возвращает любая используемая функция или ее параметры. Выделите на рабочем листе ячейку, содержащую формулу, и в строке формул выделите мышью интересующую вас часть этой формулы. Нажмите клавишу <F9>, выделенная часть формулы заменится на вычисленные значения (массивы значений). Для возврата к обычному отображению формулы нажмите клавишу <Esc> или щелкните на кнопке «Отмена» в строке формул (кнопка с косым красным крестиком).

2. Результат табличной формулы может содержаться в одной ячейке, если результирующая функция в формуле возвращает также одно значение.

Теперь можно пояснить смысл табличной формулы поля «Строки возврата» на листе «Абонемент»:

{=МИН(ЕСЛИ((Читатель\_В=A2)\*(Дата\_возврата\_В>B2)>0;СТРОКА(Читатель\_В)))}

По данной формуле если на листе «Возврат книг» нужен читатель (точнее номер его билета, заданный в ячейке A2) вернул книги позже данной даты выдачи (B2), то определяются номера таких строк, а из этих номеров ищется минимальный, иначе – значение 0). Проверьте, какие результирующие значения содержат отдель-

Таблица 1

Имя массива ячеек	Формула
Дата_возврата_А	=Абонемент!\$C\$1:\$C\$65
Дата_возврата_В	=Возврат книг !\$B\$1:\$B\$65
Дата_выдачи	=Абонемент!\$B\$1:\$B\$65
Инв_номер	=Фонд!\$A\$1:\$A\$65
Книга1	=Абонемент!\$D\$1:\$D\$65
Книга2	=Абонемент!\$E\$1:\$E\$65
Книга3	=Абонемент!\$F\$1:\$F\$65
Книги_В	=Абонемент!\$D\$1:\$F\$65
Стоимость	=Фонд!\$G\$1:\$G\$65
Строка_возврата	=Абонемент!\$K\$1:\$K\$65
Фонд	=Фонд!\$A\$1:\$G\$65
Читатель_А	=Абонемент!\$A\$1:\$A\$65
Читатель_В	=Возврат книг !\$A\$1:\$A\$65

ные сегменты формулы (для этого и выбирались укороченные массивы при задании имен).

Чтобы скрыть отображение нулевых значений в столбце, выделите ячейки столбца в меню **Формат** → **Ячейки** на вкладке «Число», в списке «Числовые форматы» выберите пункт (все форматы), а в поле Тип введите **0;-0;;@**.

Для поля «Строка абонемент» на листе «Фонд» ячейки (например, I2) содержат табличную формулу

{=МАКС(((Книга1=A2)+(Книга2=A2)+(Книга3=A2))\*(Строка\_возврата=0)\*СТРОКА(Читатель\_В))}

По этой формуле для каждой книги фонда ищутся на листе «Абонемент» все строки, в которых присутствует данная книга в качестве одной из выданных (то есть в строке либо (Книга1=A2), либо (Книга2=A2), либо (Книга3=A2) имеет значение ИСТИНА), а строка возврата – пустая. Выбирается максимальная такая строка в качестве результирующей (так как строки листа «Абонемент» заполняются по возрастанию даты выдачи книг). Примените форматирование для скрытия нулевых значений.

Модель готова. Готовы ли Вы найти для нее практическое применение и совершенствование?




Рис. 8



### ЗАДАНИЯ ДЛЯ САМОСТОЯТЕЛЬНОЙ РАБОТЫ

Выполните следующие задания.

1. Определите посещаемость библиотеки (суммарное количество читателей, которым выдавались книги) за отчетный период времени.
2. Определите «рейтинг популярности» книг фонда (сколько раз книга была выдана читателям).
3. Определите рейтинг активности читателей по количеству прочитанных ими книг, а также общее количество книг, выданных за отчетный период.

*Указания:*

Выделить данные для нужного интервала времени можно с использованием команды **Данные → Фильтр → Автофильтр**, при этом наложить должные условия на поля с датами.

1. Для подсчета количества ячеек с данными в интервале или массиве можно применить функцию СЧЁТЗ.
2. Частоту использования книги с инвентарным номером 3 (ячейка Фонд!A2)

можно определить по формуле  

$$\{=СУММ((Книга1=A2)+(Книга2=A2)+(Книга3=A2))\}$$

3. Для выполнения задания 3 подойдет в качестве нахождения промежуточных данных процедура консолидации данных (команда меню **Данные → Консолидация**). Выберите любую свободную ячейку листа «Служебный», в окне команды **Консолидация** задайте параметры, как на рис. 8, и нажмите кнопку «ОК».

Для каждого читателя создается таблица количества единиц взятых книг (по полю Книга1, Книга2, Книга3), просуммировав данные можно получить общее количество книг как для отдельного читателя, так и в совокупности по библиотеке в целом.



### ТВОРЧЕСКОЕ ЗАДАНИЕ

В определенное время библиотека проводит перерегистрацию читателей. В чем ее смысл и какие действия предусматривает данная процедура? Дополните модель данной функцией.



### РЕЗЮМЕ

Профессиональное исполнение поставленной задачи, возможно, потребовало бы использование «движка» (СУБД) для работы с базами данных. Однако для демонстрации взаимосвязей между данными был использован простейший инструмент, но с «богатыми» возможностями по обработке информации различных типов. Обучение этим возможностям при решении конкретной повседневной задачи и ставилось целью данной работы.



Наши авторы, 2007  
Our authors, 2007

*Паньгин Александр Викторович,  
инженер Центра информационных  
технологий, г. Сосновый Бор.*